Practical Tools for Knowledge Representation and Nonmonotonic Reasoning

Ilkka Niemelä and Mirek Truszczyński

AAAI 2000 — Tutorial SA2

Focus

• Part 1: Stable Logic Programming (SLP) • Part 2: Smodels — implementation of an SLP system

SA2-2

Overview — part 1

• Motivation • Knowledge representation roots • Default logic primer • Logic programming roots • Stable Logic Programming (SLP) • Answer Set Programming (ASP) • Examples

SA2-3

French Phrases, Italian Soda

• Mark T. Zegareli, Dell Logic Puzzles, page 43, June 1999 • Six people sit at a round table • Each drinks a different kind of soda • Each plans to visit a different French-speaking country • Based on clues — find out:

— what each of them is drinking — where each of them is going

SA2-4 Motivation

French Phrases, Italian Soda

• The person who is planning a trip to Quebec, who drank either blueberry or lemon soda, didn’t sit in seat number one.

• Jeanne didn’t sit next to the person who enjoyed the kiwi soda.

• The person who has a plane ticket to Belgium, who sat in seat four or seat five, didn’t order the tangelo soda.

SA2-5 Motivation

How to find a good route?

• Given:

– a set of places to visit – the set of pairs of places that have direct connections • Find a closed route that takes you through each place exactly once • Hamiltonian cycle problem • Traveling salesman problem

SA2-6 Motivation

How to move blocks around?

• Find a plan!

• Blocks on the table are arranged in stacks • They need to be rearranged into different stacks • At each step

– only top block in each stack can be moved – can be placed on top of another stack or on the table

SA2-7 Motivation

Other problems

• Classical AI puzzles and problems:

– missionaries and cannibals, 15-puzzle, magic squares, ...

• Constraint satisfaction problems – propositional satisfiability • Combinatorics

– graph problems, combinatorial optimization, combinatorial structures

SA2-8 Motivation

Goal

• WANTED: An approach for modeling and solving such problems • Diverse domains • Spatial and temporal reasoning • Constraints • Incomplete information • Frame problem

SA2-9 Motivation

Solution

• Answer Set Programming (ASP) paradigm • Stable Logic Programming (SLP) — its implementation

SA2-10 Motivation

ASP — where does it come from?

• Knowledge Representation – Commonsense reasoning – Defeasible reasoning • Declarative programming – Logic programming

SA2-11 Roots

Roots — KR

• Commonsense reasoning • Defeasible inferences • Elaboration tolerance • Representing the frame problem • Problems with FOL

– monotonicity

SA2-12 Roots — KR

Roots — KR

• Nonmonotonic logics – Default logic; Reiter 1980 – Autoepistemic logic; Moore 1984 – Circumscription; McCarthy 1980

SA2-13 Roots — KR

Default logic

• Extension of FOL by defeasible inference rules — defaults: α: β_1, . . . , β_k γ • If

– α is known and – if it is known that all β_is are possible then γ is known • Default theory: (D, W )

SA2-14 DL primer

Default logic — semantics

• Assume what’s possible

– that is, pick a theory S – β is possible if S ⊬ ¬β • default α : β γ yields S-applicable rule α γ

SA2-15 DL primer

Default logic — semantics

• Default proof — ⊢_{Δ,S} • Uses: – W – tautologies – modus ponens – S-applicable rules • S is an extension of a default theory Δ if S = {φ: ⊢_{Δ,S} φ} SA2-16 DL primer

Default logic — example • W = {p} D = { p : ¬q r , p : ¬r q , r : ¬s s }

• Exactly one extension: S = Cn(p, q)

• Default proofs w.r.t. S use: p, p q and r s • Exactly the consequences of p and q can be derived!

SA2-17 DL primer

Default logic — example • W = {p} D = { p : ¬q r , p : ¬r q , r : ¬s s }

• S = Cn(p, r, s) is NOT an extension • Default proofs w.r.t. S use p, and p r • No way to derive s!

SA2-18 DL primer

Default logic — example

• Modeling inertia (change requires an action) in location(X, L, T ) : ¬abnormal(X, T ) in location(X, L, T + 1) move(X, L1, L2, T ) ⇒ in location(X, L2, T + 1) move(X, L1, L2, T ) ⇒ abnormal(X, T )

SA2-19 DL primer

Default logic — example

• Modeling inertia (change requires an action) in location(X, L, T ) : ¬abnormal(X, T ) in location(X, L, T + 1) move(X, L1, L2, T ) ⇒ in location(X, L2, T + 1) move(X, L1, L2, T ) ⇒ abnormal(X, T ) destroyed(X, L, T ) ⇒ abnormal(X, T )

SA2-20 DL primer

Importance of default logic

• Generalizes Closed World Assumption • Generalizes logic programming with stable model semantics • Captures propositional circumscription • Captures an important fragment of

autoepistemic logic

SA2-21 DL primer

Roots — declarative programming

• Algorithm = Logic + Control (Kowalski, 1979) • First-order logic as a programming language • Expectations, hopes

– easy programming, fast prototyping – handle on program verification – advancement of software engineering

SA2-22 Roots — LP

Horn logic programming • p( ⃗X) ← q_1( ⃗X), . . . , q_k( ⃗X) • Herbrand models • single intended Herbrand model • least model

SA2-23 Roots — LP

Computing with Horn programs

• A Horn program P specifies a subset X of the Herbrand universe for P , HU(P ), if for some predicate symbol p occurring in P we have: X = {t ∈ HU(P ): p(t) ∈ LM(P )}. • Finite Horn programs specify precisely the r.e.

sets and relations • Smullyan, 1968, Andreka and Nemeti, 1978

SA2-24 Roots — LP

Computing with Horn programs

• Function symbols necessary!

• List constructor [·|·] used to define higher-order objects • Terms - basic data structures • Unification - basic mechanism to manipulate data structures • SLD-resolution • Extensive use of recursion

SA2-25 Roots — LP

Example

append([],X,X).
append([X|Y],Z,[X|T]) :- append(Y,Z,T).

reverse([],[]).
reverse([X|Y],Z) :-
reverse(Y,U),append(U,[X],Z).

• both relations defined recursively

• terms represent complex objects: lists, sets, ...

SA2-26 Roots — LP

Negation in the body

• Things get more complex! • Clauses of the form: p( ⃗X) ← q_1( ⃗X), . . . , q_k( ⃗X), not(r_1( ⃗X)), . . . , not(r_l( ⃗X)) • Why negation?

– Natural linguistic concept – Facilitates declarative descriptions

(definitions)

SA2-27 Roots — LP

Semantics of programs with negation

• Usually — no least Herbrand model! • Choosing a single intended model difficult • Great logic programming schism

– single intended model approach – multiple preferred model approach • Nonmonotonicity — the roots converge!

SA2-28 Roots — LP

Single intended model approach

• Extensions of Horn logic programming – Perfect semantics of stratified programs – 3-valued well-founded semantics for (arbitrary) programs • Top-down computing based on unification and resolution • XSB – Warren’s group, SUNY Stony Brook

SA2-29 Roots — LP

Were the expectations met?

• Logic programming (single intended model approach) – Requires significant logical background – Terms hard to think of as data structures – Extensive (often not intuitive) use of recursion – Unification difficult

SA2-30 Roots — LP

Multiple preferred models

• Attractive alternative • Semantics assigns to a program (theory) not one but many intended models! – for instance, all stable models • How to interpret these semantics?

– skeptical reasoning — resolution-based approaches too complex – preferred models represent different solutions to a problem

SA2-31 Roots — LP

Stable model semantics

• Propositional case first! • P — logic program • M — set of atoms • Reduct P^M (Gelfond, Lifschitz) – for each a ∈ M remove rules with not(a) in the body – remove literals not(a) from all other rules

SA2-32 SLP

Stable model semantics

• M-applicable clauses • Reduct P^M is a Horn program • It has the least model LM(P^M ) • M is a stable model of P if M = LM(P^M ) • Connection to default logic p ← q, not(s) → q : ¬s p SA2-33 SLP

Stable models — example

• p ← q, not(r) → p :- q, not r.

p :- q, not s.
r :- p, not q , not s.
s :- not q.
q :- not s.

• Two stable models: M_1 ={p,q} and M_2 ={s}

SA2-34 SLP

Stable models — example

• The case of M_1 ={p,q} • P_{M1} :

p :- q.
q.

• LM(P_{M1} ) ={p,q}= M_1

SA2-35 SLP

Connection to DL — example • W = {p} D = { p : ¬q r , p : ¬r q , r : ¬s s }

• Corresponding program:

p.
r :- p, not q.
q :- p, not r.
s :- r, not s.

• {p,q} is the only stable model

SA2-36 SLP

“Killing” clauses — constraints

• p :- not p.

• This program has NO stable models • P — program; p — new atom • Adding

p :- not p.

to P “kills” all stable models of P

SA2-37 SLP

“Killing” clauses — constraints

• Adding

p :- q, not r, not p.

to P “kills” all stable models of P that: – contain q, and – do not contain r • Abbreviation:

:- q, not r.

SA2-38 SLP

Programs with variables

• Herbrand models only!

• Program clause — a shorthand for all its ground substitutions

edge(1,2).
edge(1,3).
edge(2,4).
path(X,Y) :- edge(X,Y).
path(X,Y) :- edge(X,Z), path(Z,Y).

SA2-39 SLP

Programs with variables

• Clause path(X,Y) :- edge(X,Y). represents:

path(1,1) :- edge(1,1).
path(1,2) :- edge(1,2).
path(2,1) :- edge(2,1).
path(2,2) :- edge(2,2).
path(1,3) :- edge(1,3).
...

• Grounding

SA2-40 SLP

Programs with variables

• Grounding yields:

– a propositional logic program – built of atoms from the Herbrand base of P , HB(P ) – denoted ground(P ) • M ⊆ HB(P ) is a stable model of P if M is a stable model of ground(P )

SA2-41 SLP

Computing with stable semantics

• A logic program P defines:

SS(P ) = {M : M stable model of P } • This is a family of subsets of HB(P ) (Herbrand base for P ) • SS(P ) may be “large” and complex • Sets in SS(P ) are, in general, infinite and complex • Restrict syntax — eliminate function symbols

SA2-42 SLP

Problem solving with SLP

• Encode the problem as a logic program • So that stable models represent solutions • Find one (or a specified) number of stable models • Reconstruct solution(s)

SA2-43 SLP

SLP - how to program?

• Rudiments of methodology – Generate-and-test programming:

∗ generate (possible structures) ∗ weed out (unwanted ones) ∗ by adding constraints (“killing” clauses) – Separate data from program (DATALOG^¬)

SA2-44 SLP

Separating data from program

• Data:

– a finite collection of ground atoms – extensional database (edb) • Program:

– a finite collection of clauses – intensional database (idb)

SA2-45 SLP

Problem-specific approach

• Requires deep understanding of the problem • Exploits the global character of the semantics • May result in more efficient programs

SA2-46 SLP

SLP — expressivity

• (Schlipf, 1994) Given an NP-problem P, there are:

– a finite program P , and – a representation scheme edb_P of problem instances so that for every instance I of P: – edb_P (I) has size polynomial in size(I) – I has a solution if and only if P ∪ edb_P (I) has a stable model.

SA2-47 SLP

Example — graph colorability • Data (edb) – Ground atoms vtx(a) — define vertices – Ground atoms edge(a,b) — define edges • Assignment of colors

clrd(X,r) :- vtx(X), not clrd(X,b),
                     not clrd(X,g).
clrd(X,b) :- vtx(X), not clrd(X,r),
                     not clrd(X,g).
clrd(X,g) :- vtx(X), not clrd(X,r),
                     not clrd(X,b).

SA2-48 SLP

Example — graph colorability

• Imposing constraints: colorability condition

f :- clr(C), edge(X,Y), clrd(X,C), clrd(Y,C), not f.

• Colorability condition — differently

:- clr(C), edge(X,Y), clrd(X,C), clrd(Y,C).

SA2-49 SLP

Answer-set programming — ASP

• L — formal language – e.g., language of predicate logic • Formal system F in L – syntax – semantics • Semantics – T ⊆ L → Sem(T ) ⊆ 2^{HB(T )} – HB(T ) — Herbrand base for T

SA2-50 ASP

ASP — examples

• Stable Logic Programming • LP with other semantics • Classical logic • CSP • Circumscription • Disjunctive logic programming (answer sets) • DATALOG with constraints • Default logic

SA2-51 ASP

What is ASP good for?

• Search problems! • Search problem Π:

– a set of finite instances, D_Π – for each instance I ∈ D_Π:

∗ a finite set S_Π(I) of all solutions of Π

SA2-52 ASP

Solving search problem with ASP

• An algorithm A solves a search problem Π if:

– for each I ∈ D_Π, A returns the answer “NO” when S_Π(I) = ∅, and – any solution s ∈ S_Π(I), otherwise • Decision problems can be cast as special search problems

SA2-53 ASP

Solving search problems with ASP

• For a search problem Π, we need:

– an effective encoding of search problem instances as collections of ground atoms — edb_Π – a finite (instance independent) theory in F encoding the search problem — T_Π – so that solutions to Π for an instance I can be reconstructed from models of edb_Π(I) ∪ T_Π

SA2-54 ASP

Solving search problems with ASP

• Formally, we require that:

– Sem(edb_Π(I) ∪ T_Π) = ∅ iff S_Π(I) = ∅, and – there is a polynomially computable onto function sol_Π that for every I ∈ D_Π maps elements of Sem(edb_Π(I) ∪ T_Π) onto S_Π(I)

SA2-55 ASP

Solving search problems with ASP

• Π can be solved for an instance I by:

– constructing the theory edb_Π(I) ∪ T_Π – finding one of its models s ∈ Sem(edb_Π(I) ∪ T_Π) – reconstructing from s a solution sol^I_Π(s) • Uniform solution for Π • Problem-specific approaches possible but harder

SA2-56 ASP

ASP - desiderata

• Expressive power • Ease of modeling – intuitive semantics – concise encodings – modular programming • Performance (speed)

SA2-57 ASP

How do they stack up?

• SLP vs PL vs DLP vs DC? • No clear answer • More experimental work needed • Nonmonotonicity of the underlying system seems to be advantageous: – conciseness – modeling (knowledge representation)

SA2-58 ASP

French Phrases, Italian Soda

• The person who is planning a trip to Quebec, who drank either blueberry or lemon soda, didn’t sit in seat number one

:- position_visits(1,quebec) .
soda_visits(blueberry,quebec) :-
               not soda_visits(lemon,quebec) .
soda_visits(lemon,quebec) :-
               not soda_visits(blueberry,quebec) .

SA2-59 Examples

French Phrases, Italian Soda

• Jeanne didn’t sit next to the person who enjoyed the kiwi soda

:- position(Position1), position(Position2),
   person_position(jeanne,Position1),
   position_soda(Position2,kiwi),
   differ_position_1(Position1,Position2) .

SA2-60 Examples

French Phrases, Italian Soda

• Closure rules

person_soda(Person,Soda) :-
       person(Person), soda(Soda), visits(Visits),
       person_visits(Person,Visits),
       soda_visits(Soda,Visits) .

SA2-61 Examples

Overview — part 2

• Smodels— introduction • Grounding — lparse • Searching — smodels • Using Smodels— examples:

– propositional satisfiability, Hamiltonian cycles, blocks world planning • Applications • Conclusions

SA2-62

Implementing SLP

• Typically a two phase technique employed – Grounding:

Given program P with variables, generate a set of ground instances of the rules (a subset of ground(P )) which preserves the stable models.

– Stable models for ground programs:

Davis-Putnam type backtracking search • Used in: Smodels, dlv, dcs, ...

SA2-63 Smodels— intro

Smodels

• Normal logic programs with extensions • Helsinki University of Technology • Developers: Ilkka Niemelä, Patrik Simons, and Tommi Syrjänen • http://www.tcs.hut.fi/Software/smodels/ • Implemented in C++ • Two modules:

– lparse (grounding) – smodels (stable models for ground programs)

SA2-64 Smodels— intro

Rules with variables

• Function symbols are not allowed (decidability) • Built-in (e.g. arithmetic) functions supported • A rule with variables is mapped to a set of ground rules by – evaluating first “easy” predicates

(range/domain predicates for variables) and – generating for the remaining predicates ground instances compatible with the “easy” ones

SA2-65 Grounding — lparse

Grounding

• Suppose that clr(C) and edge(X,Y) are “easy” and have been evaluated • Then for a rule,

:- clr(C), edge(X,Y), clrd(X,C), clrd(Y,C).

grounding produces all ground rules

:- clr(r), edge(a,b), clrd(a,r), clrd(b,r).

for which clr(r) and edge(a,b) hold

SA2-66 Grounding — lparse

Grounding — cont’d

• “Easy” predicates:

– in Smodels: non-recursive (database views) • Consider rules:

e2(X,Y) :- edge(X,Z), edge(Z,Y), not edge(X,Y).
d2(Y) :- e2(X,Y).

• If edge(X,Y) is an “easy” predicate, then so are

e2(X,Y), d2(Y)

SA2-67 Grounding — lparse

Grounding — cont’d

• For safety: domain-restricted programs • Each variable has a domain (“easy”) predicate

clrd(X,r) :- vtx(X),
        not clrd(X,b), not clrd(X,g).

• Straightforward to incorporate built-in functions, e.g., arithmetic (no floundering problems)

even(X+1) :- d(X), not even(X).

SA2-68 Grounding — lparse

Grounding — cont’d

• For domain-restricted programs, grounding can be implemented locally and efficiently using database techniques • Stronger notions of “easy” predicates possible • In dlv:

– range-restricted programs – “easy” predicates evaluated using deductive database techniques

SA2-69 Grounding — lparse

Search in smodels

• Stable model computation:

– Bottom-up backtracking search building a binary search tree – choice points on atoms:

SA2-70 Searching — smodels

Search in smodels

• To prune the search space:

– choices are made one at a time (heuristic) – before making a new choice, the set of remaining choice points is pruned and conflicts are detected (propagation rules) • Low level implementation using Dowling-Gallier type data structures

SA2-71 Searching — smodels

Propagation rules

• Similar to unit propagation:

Given q ← p_1, not(p_2) and {p_1, not(p_2)} derive q Given q ← p_1, not(p_2) and {not(q), p_1} derive p_2 • In addition, some stronger rules are used:

Given only one rule for q: q ← p_1, not(p_2) and {q} derive p_1, not(p_2).

SA2-72 Searching — smodels

Propagation rules — cont’d

Rule P A Derived Fitting-1 q ← p_1, not(p_2) p_1, not(p_2) q Fitting-2 q ← p_1, not(p_2) p_2, not(p_3) not(q) q ← p_3, not(p_4)

SA2-73 Searching — smodels

Propagation rules — cont’d

Rule P A Derived BP-1 q ← p_1, not(p_2) q p_1, not(p_2) BP-2 q ← p_1, not(p_2) not(q), p_1 p_2 Upper Bound p_2 ← p_1, not(p_3) − not(p_1), p_3 ← p_2, not(p_1) not(p_2), p_1 ← p_3, not(p_2) not(p_3)

SA2-74 Searching — smodels

Propagation rules — cont’d

• BP-1 and Upper bound have no counter part in SAT checkers • Denote by expand(P, A) the closure of A under all the rules (Fitting-1 — Upper Bound) • expand(P, A) generalizes the well-founded model • Specifically: expand(P, ∅) is the well-founded model of P

SA2-75 Searching — smodels

Propagation rules — cont’d

• expand(P, A) can be computed in quadratic time:

– closure of (Fitting-1 – BP-2) in linear time – Upper Bound in linear time • expand is the key procedure in Smodels (but e.g. dlv does not use Upper bound)

SA2-76 Searching — smodels

Lookahead

• Smodels uses an even stronger propagation rule • Lookahead rule

– Given program P and literals A:

∗ if for a literal L, expand(P, A ∪ {L}) contains a conflict (p, not(p)) ∗ then derive the complement of L

SA2-77 Searching — smodels

Lookahead — cont’d

• Lookahead can yield a smaller search space:

SA2-78 Searching — smodels

Search heuristics

• Typical approaches to selecting choice points:

– Maximum Occurrences in rules of Minimum size (MoM’s) heuristics – Unit propagation counting – Preferring choice points with the smallest combined remaining search space (Smodels)

SA2-79 Searching — smodels

Heuristics in smodels

• Computed while doing lookahead • For each undetermined atom p, two lookaheads: A_p = expand(P, A ∪ {p}) A_{not(p)} = expand(P, A ∪ {not(p)})

SA2-80 Searching — smodels

Heuristics in smodels — cont’d

• If p is the next choice point, an upper bound on the remaining search space is E(p) = 2^u + 2^{u′} where u (u^′) is the number of undetermined atoms in A_p (A_{not(p)}) • Heuristic: choose an atom p with the smallest estimated remaining search space E(p)

SA2-81 Searching — smodels

Further extensions

• Choices over subsets • { a1,..., an } :- b, not c.

– If the body holds, then any subset of {a1,...,an} (including empty) can be derived • Choice rules

SA2-82 Extensions — Smodels

Choice rules — an example

• Rule { a } :- b.

– Corresponds to a :- not na, b.

na :- not a.

– One stable model: {} • Program: { a } :- b.

b.

– Two stable models: {b}, {b,a}

SA2-83 Extensions — Smodels

Further extensions — cont’d

• Cardinality constraints • l { p_1, . . . , p_n } u – satisfied in a model S if the number of the literals from {p_1, . . . , p_n} satisfied in S is between integers l and u (inclusive) • Example: vertex cover of size less than 12:

1 { vi,vj } 2 . % for all edges (vi,vj)
:- 12 { v1,...,vn } .

SA2-84 Extensions — Smodels

Further extensions — cont’d

• Weight constraints • l ≤ {p_1 = w_1, . . . p_n = w_n} ≤ u – satisfied in S if the sum of the weights w_i of the literals from {p_1, . . . , p_n} satisfied in S is between integers l and u (inclusive)

SA2-85 Extensions — Smodels

Further extensions — cont’d

• Example: Knapsack with the sum of weights at most w but the sum of the values exceeding v

0 [ i1=w1,...,in=wn ] w.
:- [ i1=v1,...,in=vn ] v.

• In the ground case, complexity still in NP

SA2-86 Extensions — Smodels

Further extensions — cont’d

• Optimization: minimize { p1=w1,...,pn=wn } . • Example: Cardinality minimal model

minimize { a1=1,...,an=1 }.

SA2-87 Extensions — Smodels

Further Extensions — cont’d

• Type definitions (conditions):

1 { clrd(V,C):clr(C) } 1 :- vtx(V).

• Two-phase implementation of the extensions:

– Grounding (lparse) handles

type definitions/conditions – Stable model computation for ground programs (smodels) extended to support directly cardinality and weight constraints.

SA2-88 Extensions — Smodels

Running Smodels

• Typical Unix command line:

UNIX> lparse [options] files | smodels [number]

• Some options:

-c const=n : define ’const’ as number ’n’
-d : domain predicates to be output
-t : output plain text
number determines the number of stable
   models computed (0 indicates all)

SA2-89 Using Smodels

Running Smodels — cont’d

UNIX> lparse demo.lp | smodels 0
smodels version 2.25. Reading...done
Answer: 1
Stable Model: r(a) p(a) s(a,b,c)
Answer: 2
Stable Model: r(a) q(a) s(a,b,c)
False demo.lp:

                                        s(a,b,c).
                                        s(X,Y,Z):- t(Y,X,Z).
                                        p(X):- not q(X), s(X,Y,Z).
                                        q(X):- not p(X), s(X,Y,Z).
                                        r(X):- p(X), s(X,Y,Z).
                                        r(X):- q(X), s(X,Y,Z).

UNIX> lparse -d none -t demo.lp
r(a) :- p(a).
r(a) :- q(a).
q(a) :- not p(a).
p(a) :- not q(a).

SA2-90 Using Smodels

Using answer-set programming

• A basic methodology: generate and test – Generator: provides candidate answer sets (typically encoded using odd loops/disjunctions/choice constructs) – Tester: eliminates non-valid candidates (typically encoded using killing clauses/integrity constraints)

SA2-91 Using Smodels

Generate-and-test

• Integrity constraints • :- a1,..., an, not b1,..., not bm. • eliminate stable models but cannot introduce new ones:

– Let P be a program and IC a set of integrity constraints – Then S is a stable model of P ∪ IC iff:

∗ S is a stable model of P , and ∗ S satisfies all ICs

SA2-92 Using Smodels

Example: satisfiability

• Generator – Candidate answer sets: propositional models, i.e., sets of atoms – For each atom ai, rule { ai }. – For program

{ a1 }.

. . .

{ an }.

2^n stable models:

{ }

. . .

{ a1,...,an }

SA2-93 Using Smodels

Satisfiability — cont’d • Tester for a1 ∨ · · · ∨ an ∨ ¬b1 ∨ · · · ∨ ¬bm :

:- not a1,..., not an, b1,..., bm.

• Clauses T a ∨ ¬b ¬b ∨ ¬a b ∨ a

Program P_T

:- not a, b.
:- a, b.
:- not a, not b.
{ a }. { b }.

Stable model

{ a }

• Models of T and stable models of P_T correspond

SA2-94 Using Smodels

Satisfiability — cont’d

• Consider general formulae (not in CNF): (a ∨ ¬b) ∧ (¬a ↔ b) • Generator (for each atom x, rule { x }. ):

{ a }.
{ b }.

SA2-95 Using Smodels

Satisfiability — cont’d

• Tester — evaluates a formula q recursively • For each subformula:

– the conditions under which it is true are given – false cases by default: it is false unless otherwise stated • A satisfying truth assignment: a stable model satisfying

:- not q.

SA2-96 Using Smodels

Satisfiability — cont’d

• Tester encoding

Subformula p Rules l_1 ∧ · · · ∧ l_n p ← p_{l1} , . . . , p_{ln} l_1 ∨ · · · ∨ l_n p ← p_{l1} . . . p ← p_{ln} ¬l p ← not(p_l) l_1 ↔ l_2 p ← p_{l1} , p_{l2} p ← not(p_{l1} ), not(p_{l2} )

SA2-97 Using Smodels

Satisfiability — cont’d

• Consider formula p_1: (a ∨ ¬b)

p_2 ∧ (¬a ↔ b) p_3

• Program: { a }. { b }.

:- not p1. p1:- p2, p3.
p2:- a. p3:- a, not b.
p2:- not b. p3:- not a, b.

Stable model:

{a,p1,p2,p3}

• Satisfying truth assignments for p_1 and the stable models of the program correspond

SA2-98 Using Smodels

Experimental results — DES

• DES (Boolean function) encoded directly as rules

Plaintext (64 bits) Key (56 bits) ↘ ↙ DES (Boolean function) ⇓ Ciphertext (64 bits)

SA2-99 Using Smodels

DES — cont’d

• Experimenting with key search problems • Known plaintext attack:

given plaintext and ciphertext find a possible key for a limited version of DES (1–3 rounds) • Smodels handles 3 rounds

SA2-100 Using Smodels

DES — cont’d

• State-of-the art SAT-checkers

(SATO/ntab/walksat) using an optimized encoding: 2 rounds • A tuned SAT-checker (rel sat based) using an optimized encoding: 3 rounds

SA2-101 Using Smodels

Example — Hamiltonian cycles

• A Hamiltonian cycle: a closed path that visits all vertices of the graph exactly once • Input: a graph

– vtx(a),... – edge(a,b),...

– initialvtx(a0), for some vertex a0

SA2-102 Using Smodels

Hamiltonian cycles — cont’d

• Candidate answer sets: subsets of edges • Generator:

{ hc(X,Y) } :- edge(X,Y).

• Stable models of the generator given a graph:

– input graph + – a subset of the ground facts hc(a,b) for which there is an input fact edge(a,b).

SA2-103 Using Smodels

Hamiltonian cycles — cont’d

• Tester (i):

Each vertex has at most one chosen incoming edge and one outgoing edge

:-hc(X,Y), hc(X,Z), edge(X,Y), edge(X,Z), Y!=Z.
:-hc(Y,X), hc(Z,X), edge(Y,X), edge(Z,X), Y!=Z.

• Only subsets of chosen edges hc(v,u) forming paths (possibly closed) pass the test

SA2-104 Using Smodels

Hamiltonian cycles — cont’d

• Tester (ii):

Every vertex is reachable from a given initial vertex through chosen hc(v,u) edges:

:- vtx(X), not r(X).
r(Y) :- hc(X,Y), edge(X,Y), initialvtx(X).
r(Y) :- hc(X,Y), edge(X,Y), r(X),
        not initialvtx(X).

• Only Hamiltonian cycles pass the tests (i–ii)

SA2-105 Using Smodels

Hamiltonian cycles — cont’d

• Given:

– the graph, the generator rule, and the tester rules (i–ii) Hamiltonian cycles and stable models correspond • A cycle: facts hc(v,u) in a stable model • Note that Hamiltonian cycles are challenging for SAT solvers as no compact SAT encodings (with linear number of atoms) are known

SA2-106 Using Smodels

Hamiltonian cycles — cont’d

• Cardinality constraints enable an even more compact encoding • Tester (i) using 2 variables:

:- 2 { hc(X,Y):edge(X,Y) }, vtx(X).
:- 2 { hc(X,Y):edge(X,Y) }, vtx(Y).

SA2-107 Using Smodels

Example: planning

• Given:

– a set of operators – initial situation and goal • find a sequence of operator instances leading from initial to goal situation

SA2-108 Using Smodels

Planning — cont’d

• Planning is PSPACE-complete • Planning with:

– deterministic operators – complete knowledge about the initial situation, and with – an upper bound on the length of the plan is NP-complete

SA2-109 Using Smodels

Block-world planning

(operator moveop
  (params (<X> OBJECT) (<Y> OBJECT) )
  (preconds (clear <X>) (clear <Y>))
  (effects (on <X> <Y>) (clear <X>)))

a b c initial situation b c a goal solution:

moveop(a,table,0), moveop(c,a,1), moveop(b,c,2)

SA2-110 Using Smodels

Mapping planning to rules

• Devise a logic program such that stable models correspond to plans: – of length at most n – that are valid – and that reach the goal

SA2-111 Using Smodels

Mapping planning to rules

• Candidate answer sets: valid execution sequences (of length ≤ n) of operator instances from the initial conditions • Tester: eliminates those sequences that do not reach the goal

SA2-112 Using Smodels

Planning — cont’d

• Preliminaries – Add to each predicate a situation argument – on(X,Y,T): X is on Y in T – moveop(X,Y,T): X is moved onto Y in T – Length bound n: time(0..n).

– nextstate(Y,X) :- time(X), time(Y),
          Y = X + 1.

SA2-113 Using Smodels

Planning — cont’d

• Available blocks: block(a).

block(b).
block(c).

• Initial conditions: on(a,b,0).

on(b,table,0).
on(c,table,0).

SA2-114 Using Smodels

Planning — cont’d

• Auxiliary concepts make encoding easier

• Rules make it straightforward to define auxiliary predicates:

object(table).
object(X) :- block(X).
covered(X,T) :- block(Z), block(X), time(T),
        on(Z,X,T).

SA2-115 Using Smodels

Planning — cont’d

• Further predicates:

on_something(X,T) :-
        block(X), object(Z), time(T),
        on(X,Z,T).
available(table,T) :- time(T).
available(X,T) :- block(X), time(T),
        on_something(X,T).

SA2-116 Using Smodels

Planning — cont’d

• Generator: execution sequences of operators

• An operator can be applied if preconditions hold:

{ moveop(X,Y,T) }:-
        time(T), block(X), object(Y),
        X != Y, on_something(X,T),
        available(Y,T),
        not covered(X,T),
        not covered(Y,T).

SA2-117 Using Smodels

Planning — cont’d

• Operator effects:

on(X,Y,T2) :- block(X), object(Y),
        nextstate(T2,T1),
        moveop(X,Y,T1).

SA2-118 Using Smodels

Planning — cont’d

• Frame axioms (as rules with exceptions):

on(X,Y,T2) :- block(X), object(Y),
        nextstate(T2,T1),
        on(X,Y,T1),
        not moving(X,T1).
% the exceptions
moving(X,T) :- time(T), block(X), object(Y),
        moveop(X,Y,T).

SA2-119 Using Smodels

Planning — cont’d

• In addition rules for blocking conflicting operator instances are needed • This set depends on how much concurrency in the search of a plan is allowed • Computationally advantageous to allow concurrency to decrease search space explosion due to interleavings of independent operators

SA2-120 Using Smodels

Planning — cont’d

• Blocking conditions for moveop (no concurrent actions):

:- 2 { moveop(X,Y,T):block(X):object(Y) },
       time(T).

SA2-121 Using Smodels

Planning — cont’d

• Blocking conditions for moveop (with concurrent actions) I–II:

% A block cannot be moved to two destination
:- 2 { moveop(X,Y,T):object(Y) },
       block(X), time(T).
% The destination cannot be moving
:- block(X), object(Y), time(T),
        moveop(X,Y,T),
        moving(Y,T).

SA2-122 Using Smodels

Planning — cont’d

• Blocking conditions for moveop (with concurrent actions) III:

% No two blocks moved onto the same block
:- 2 { moveop(X,Y,T):block(X) },
       block(Y), time(T).

SA2-123 Using Smodels

Planning — cont’d

• Tester: excludes models where the goal has not been reached

:- not goal.
goal :- time(T), goal(T).
goal(T2) :- nextstate(T2,T1), goal(T1).
% Actual goal conditions
goal(T) :- time(T),
        on(b,c,T),
        on(c,a,T).

SA2-124 Using Smodels

Planning — cont’d

• Plans correspond to stable models:

– there is a stable model iff there is a valid sequence of moves that leads to goal and can be executed concurrently in at most n steps • A valid plan

– facts moveop(x,y,t) in a model

ordered by the argument t where facts with the same t can be taken in any linear order.

SA2-125 Using Smodels

Planning — cont’d

• Easy to add optimizations:

% Stop when the goal has been reached
:- block(X), object(Y), time(T),
        moveop(X,Y,T),
        goal(T).

SA2-126 Using Smodels

Planning — cont’d

• Further optimizations (pruning rules):

% No move from table to table
:- block(X), time(T),
       moveop(X,table,T), on(X,table,T).

% No move on something and then to table
:- nextstate(T2,T1), block(X), object(Y),
        moveop(X,Y,T1), moveop(X,table,T2).

SA2-127 Using Smodels

Planning — cont’d

• Encoding with card. constraints (Ext) is more compact and efficient than that using normal rules (Nor).

Blocks world n Ground rules Time (s) problem Nor/Ext Nor/Ext large.c 8 81 681/19 311 first plan 9.2/2.3 (15 blocks) 7 72 527/17 087 no plan 1.4/0.4 large.d 9 127 999/27 359 first plan 21/4.4 (17 blocks) 8 115 109/24 533 no plan 3.1/0.6 large.e 10 191 621/37 379 first plan 40/6.7 (19 blocks) 9 174 099/33 879 no plan 5.0/0.8

SA2-128 Using Smodels

Future work

• Improving implementation • Development tools (debuggers, tracers) • Integration with other systems and tools (SAT, CSP, Databases, programming languages) • Programming methodology

SA2-129 Using Smodels

Applications

• Planning

Dimopoulos, Nebel, Koehler: Encoding Planning Problems in Nonmonotonic Logic Programs Lifschitz: Answer Set Planning • Model checking Liu, Ramakrishnan, and Smolka: Fully Local and Efficient Evaluation of Alternating Fixed Points

SA2-130 Applications

Applications — cont’d

• Reachability analysis Heljanko: Using Logic Programs with Stable Model Semantics to Solve Deadlock and Reachability Problems for 1-Safe Petri Nets See, the mcsmodels tool:

http://www.tcs.hut.fi/~kepa/tools/

Varpaaniemi: Stable Models for Stubborn Sets

SA2-131 Applications

Applications — cont’d

• Product configuration

Soininen and Niemelä: Developing a declarative rule language for applications in product configuration Syrjänen: A Linux configurator • Dynamic Constraint Satisfaction Soininen and I. Niemelä and E. Gelle: A Fixpoint Definition of Dynamic Constraint Satisfaction

SA2-132 Applications

Applications — cont’d

• Feature interaction Accorsi, Areces and Maarten: Towards Feature Interaction via Stable Models • Logical cryptanalysis Hietalahti, Massacci, and Niemelä: DES: a Challenge Problem for Nonmonotonic Reasoning Systems

SA2-133 Applications

Applications — cont’d

• Network inhibition analysis

Aura, Bishop, Sniegowski: Analyzing single-server network inhibition • Computing (total and partial) stable models for disjunctive programs Janhunen, Niemelä, Simons, and You: Unfolding Partiality and Disjunctions in Stable Model Semantics

SA2-134 Applications

Other ASP systems

• dlv (disjunctive logic programs)

http://www.dbai.tuwien.ac.at/proj/dlv/

• dcs (Datalog with constraints)

http://www.cs.engr.uky.edu/dcs/

• LDL^{++} (Stratified programs with choice constructs and aggregates)

http://www.cs.ucla.edu/ldl/

• CCALC (tight programs)

http://www.cs.utexas.edu/users/tag/cc/

SA2-135 Conclusions

Other systems — dlv

• Disjunctive logic programming • Technical University of Vienna • Nicola Leone and his group • http://www.dbai.tuwien.ac.at/proj/dlv/

SA2-136 Conclusions

Dlv • Disjunctive clauses: • Clauses of the form:

p_1| . . . |p_k ← q_1, . . . , q_m, not(r_1), . . . , not(r_n) • Colorability:

clr(X,r) | clr(X,g) | clr(X,b) :- vtx(X).

SA2-137 Conclusions

Dcs • DATALOG with constraints • University of Kentucky • Developers: Deborah East, Mirosław Truszczyński • http://www.cs.engr.uky.edu/dcs/

SA2-138 Conclusions

Dcs • Theories consist of

– Constraints and post-constraints – Horn clauses (DATALOG) • Semantics: M is a model if:

– M is a model of constraints – the closure of M under Horn rules is a model of post-constraints

SA2-139 Conclusions

Conclusions

• ASP emerges as a viable KR tool • Efficient implementations (Smodels, dlv, dcs) • Expanding functionality and ease of use • Growing range of applications • Use it!

SA2-140 Conclusions